RATIO Example #5

SUDAAN Statements and Results Illustrated

- REPWGT
- WEIGHT
- ADJFAY option
- NUMER
- DENOM

Input Data Set(s):  DESCRIPTT.XPT

Example

Obtain RATIO estimates and variances using multiply imputed data from NHANES III.

Specifically, use RATIO to estimate the mean body weight among respondents who consider their health ‘fair’ or ‘poor’, by poverty status.

Solution

This example uses the NHANES III Multiply Imputed Data Set, which is available on the NCHS website. In the following example, we illustrate how to program SUDAAN for a multiply imputed dataset when the multiply imputed variables are stored as separate variables on the same datafile.

A temporary dataset called DESCRPTT.XPT was created from the NHANES III Multiply Imputed Dataset, which contained the necessary weight and variance estimation variables as well as the following newly constructed variables:

POVERTY1—POVERTY5

Created from the multiply imputed variables DMPPIRMI and is equal to:
1  – If respondent is at or below poverty
2  – If respondent is above poverty

WEIGHT1—WEIGHT5

Created from the multiply imputed variables WEIGHTMI and is equal to:
1  – WEIGHTMI, if the respondent indicated their health was “Fair” or “Poor” based on the corresponding HEALTH variable (below)
0  – Otherwise

HEALTH1—HEALTH5

Page 1 of 7
Created from the multiply imputed variables HABLMI and is equal to:

1 – if the respondent indicated their health was “Fair” or “Poor”

0 – Otherwise

For each of the variables above, five multiply imputed versions of the variable are available on the NHANES III dataset. In this example, variances are computed using the BRR method with Fay’s Adjustment.

In the first call to PROC RATIO, the mean body weight among respondents who consider their health ‘fair’ or ‘poor’ is computed by poverty status (Exhibit 1).

This example was run in SAS-Callable SUDAAN, and the programming code is presented in Exhibit 1. Note that the basic SUDAAN code is the same for both Standalone and SAS-Callable versions.

Exhibit 1. SAS-Callable SUDAAN Code: First RATIO Run

```sas
libname in xport "\rtints29\sudaan\data\nhanes3\descrptT.xpt";
options linesize=95 pagesize=60 nocenter;
data descrptT; set in.descrptT;
proc format;
   value pov 1="1=At or Below Poverty"
             2="2=Above Poverty";
PROC RATIO DATA=descrptT filetype=sas DESIGN=BRR;
   WEIGHT WTPFQX6;
   REPWGT WTPQRP1-WTPQRP52 / ADJFAY=2.0408;
   MI_VAR POVERTY1 POVERTY2 POVERTY3 POVERTY4 POVERTY5;
   MI_VAR WEIGHT1  WEIGHT2  WEIGHT3  WEIGHT4  WEIGHT5;
   MI_VAR HEALTH1  HEALTH2  HEALTH3  HEALTH4  HEALTH5;
   CLASS POVERTY1;
   NUMER WEIGHT1;
   DENOM HEALTH1;
   SETENV LABWIDTH=25 decwidth=2 colwidth=6;
   PRINT NSUM WSUM RHAT SERHAT="SE" LOWRHAT UPRHAT / STYLE=NCHS NSUMFMT=F6.0
               WSUMFMT=F9.0 SERHATFMT=F5.2;
   RFORMAT POVERTY1 pov.;
   RTITLE "MEAN WEIGHT AMONG ADULTS WHO INDICATED THEIR HEALTH IS 'FAIR' OR 'POOR'"
             "BY POVERTY STATUS";
   RFOOTNOTE "NHANES-III MULTIPLY IMPUTED DATA, ADULTS (20+)";
```

The NUMER (weight) and DENOM (health) variables have been defined in such a way that they will estimate the average weight among people who report their health as “fair” or “poor”.
Exhibit 2. First Page of SUDAAN Output (SAS *.lst file)

| Sample Weight: WTPFQX6 |
|---|---|
| Replicate Sample Weights: |
| WTPQRP1 WTPQRP2 WTPQRP3 WTPQRP4 WTPQRP5 WTPQRP7 WTPQRP8 |
| WTPQRP9 WTPQRP10 WTPQRP11 WTPQRP12 WTPQRP13 WTPQRP14 WTPQRP15 WTPQRP16 |
| WTPQRP17 WTPQRP18 WTPQRP19 WTPQRP20 WTPQRP21 WTPQRP22 WTPQRP23 WTPQRP24 |
| WTPQRP25 WTPQRP26 WTPQRP27 WTPQRP28 WTPQRP29 WTPQRP30 WTPQRP31 WTPQRP32 |
| WTPQRP33 WTPQRP34 WTPQRP35 WTPQRP36 WTPQRP37 WTPQRP38 WTPQRP39 WTPQRP40 |
| WTPQRP41 WTPQRP42 WTPQRP43 WTPQRP44 WTPQRP45 WTPQRP46 WTPQRP47 WTPQRP48 |
| WTPQRP49 WTPQRP50 WTPQRP51 WTPQRP52 |

Multiplier Associated with Replicate Weights: 2.0408

A separate summary of file counts is provided for each imputation (Exhibit 2). There are 18,825 observations on each imputed dataset, with 52 denominator degrees of freedom.
Exhibit 3. Frequencies for CLASS Variable POVERTY1

<table>
<thead>
<tr>
<th>POVERTY1</th>
<th>Frequency</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>4446.800</td>
<td>1=At or Below Poverty</td>
</tr>
<tr>
<td>2</td>
<td>14378.200</td>
<td>2=Above Poverty</td>
</tr>
</tbody>
</table>

Exhibit 4. RATIO Results: Estimates Within Poverty Status

Variance Estimation Method: BRR Using Multiply Imputed Data

MEAN WEIGHT AMONG ADULTS WHO INDICATED THEIR HEALTH IS 'FAIR' OR 'POOR' BY POVERTY STATUS

Results for Summary Over All Imputations by: Variable, POVERTY1.

<table>
<thead>
<tr>
<th>Variable</th>
<th>Lower 95%</th>
<th>Upper 95%</th>
<th>Sample Size</th>
<th>Weighted Size</th>
<th>Ratio Est.</th>
<th>SE</th>
<th>Ratio Limit 95% Lower</th>
<th>Ratio Limit 95% Upper</th>
</tr>
</thead>
<tbody>
<tr>
<td>WEIGHT1/HEALTH1 Total</td>
<td>167.67</td>
<td>170.01</td>
<td>18825</td>
<td>177180670</td>
<td>167.67</td>
<td>1.16</td>
<td>165.34</td>
<td>170.01</td>
</tr>
<tr>
<td>1=At or Below Poverty</td>
<td>163.39</td>
<td>167.35</td>
<td>4447</td>
<td>23269645</td>
<td>163.39</td>
<td>1.97</td>
<td>159.43</td>
<td>167.35</td>
</tr>
<tr>
<td>2=Above Poverty</td>
<td>169.34</td>
<td>171.79</td>
<td>14378</td>
<td>153911025</td>
<td>169.34</td>
<td>1.22</td>
<td>166.89</td>
<td>171.79</td>
</tr>
</tbody>
</table>

Exhibit 4 displays the overall ratio estimates and associated standard errors and 95% confidence limits. This page shows that the average weight among the people with fair/poor health at or below poverty is 163.39 pounds (95% confidence limits of 159-167 lbs), and the average weight among people with fair/poor health in the above poverty group is 169.34 pounds (95% confidence limits of 167-172 lbs).

NOTE: The BY_MI option on the PRINT statement would have produced estimates for each imputation as well as overall.
In the next call to RATIO (Exhibit 5), we estimate the difference in ratios between adults at or below poverty vs. those above poverty.

**Exhibit 5. SAS-Callable SUDAAN Code: Second Call to RATIO**

```sas
PROC RATIO DATA=descrptT FILETYPE=sas DESIGN=BRR NOMARG;
  WEIGHT WTPFQX6;
  REPWGT WTPQRP1-WTPQRP52 / ADJFAY=2.0408;
  MI_VAR POVERTY1 POVERTY2 POVERTY3 POVERTY4 POVERTY5;
  MI_VAR WEIGHT1 WEIGHT2 WEIGHT3 WEIGHT4 WEIGHT5;
  MI_VAR HEALTH1 HEALTH2 HEALTH3 HEALTH4 HEALTH5;
  CLASS POVERTY1;
  NUMER WEIGHT1;
  DENOM HEALTH1;
  CONTRAST POVERTY1=(1 -1) / name="Below vs. Above Poverty";
  SETENV LABWIDTH=30 colwidth=6 decwidth=2;
  PRINT RHAT=“Difference” SERHAT=“SE” T_RHAT=“T-Stat” P_RHAT=“P-value” /
       STYLE=NCHS rhatfmt=f10.2 serhatfmt=f6.2 p_rhatfmt=f7.4;
  RTITLE "MEAN WEIGHT AMONG ADULTS WHO INDICATED THEIR HEALTH IS 'FAIR' OR 'POOR'"
       "Comparison of Adults Across Poverty Status";
  RFOOTNOTE "NHANES-III MULTIPLY IMPUTED DATA, ADULTS (20+)";
```
Exhibit 6. First Page of SUDAAN Output (SAS *.lst file)

S U D A A N
Software for the Statistical Analysis of Correlated Data
Copyright Research Triangle Institute December 2011
Release 11.0.0

DESIGN SUMMARY: Variances will be computed using the Balanced Repeated Replication (BRR) Method
Sample Weight: WTPFQX6
Replicate Sample Weights:
  WTPQRP1 WTPQRP2 WTPQRP3 WTPQRP4 WTPQRP5 WTPQRP7 WTPQRP8
  WTPQRP9 WTPQRP10 WTPQRP11 WTPQRP12 WTPQRP13 WTPQRP14 WTPQRP15 WTPQRP16
  WTPQRP17 WTPQRP18 WTPQRP19 WTPQRP20 WTPQRP21 WTPQRP22 WTPQRP23 WTPQRP24
  WTPQRP25 WTPQRP26 WTPQRP27 WTPQRP28 WTPQRP29 WTPQRP30 WTPQRP31 WTPQRP32
  WTPQRP33 WTPQRP34 WTPQRP35 WTPQRP36 WTPQRP37 WTPQRP38 WTPQRP39 WTPQRP40
  WTPQRP41 WTPQRP42 WTPQRP43 WTPQRP44 WTPQRP45 WTPQRP46 WTPQRP47 WTPQRP48
  WTPQRP49 WTPQRP50 WTPQRP51 WTPQRP52
Multiplier Associated with Replicate Weights: 2.0408

Processing data for set 1 of imputed variables:
Number of observations read : 18825 Weighted count :177180670
Denominator degrees of freedom : 52

Processing data for set 2 of imputed variables:
Number of observations read : 18825 Weighted count :177180670
Denominator degrees of freedom : 52

Processing data for set 3 of imputed variables:
Number of observations read : 18825 Weighted count :177180670
Denominator degrees of freedom : 52

Processing data for set 4 of imputed variables:
Number of observations read : 18825 Weighted count :177180670
Denominator degrees of freedom : 52

Processing data for set 5 of imputed variables:
Number of observations read : 18825 Weighted count :177180670
Denominator degrees of freedom : 52

Exhibit 7. Frequencies for CLASS Variable POVERTY1

<table>
<thead>
<tr>
<th>POVERTY1</th>
<th>Frequency</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>Ordered Position: 1</td>
<td>4446.800</td>
<td>1</td>
</tr>
<tr>
<td>Ordered Position: 2</td>
<td>14378.200</td>
<td>2</td>
</tr>
</tbody>
</table>
Exhibit 8. **RATIO Results: CONTRAST Across Poverty Status**

Variance Estimation Method: BRR Using Multiply Imputed Data

**MEAN WEIGHT AMONG ADULTS WHO INDICATED THEIR HEALTH IS 'FAIR' OR 'POOR'**

Comparison of Adults Across Poverty Status

**Results for Summary Over All Imputations**

by: Variable, SUDAAN Reserved Variable One, Contrast.

for: Variable = WEIGHT1/HEALTH1.

<table>
<thead>
<tr>
<th>SUDAAN Reserved Variable One</th>
<th>Difference</th>
<th>SE</th>
<th>T-Stat</th>
<th>P-value</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Contrast</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Below vs. Above Poverty</td>
<td>-5.95</td>
<td>2.01</td>
<td>-2.96</td>
<td>0.0047</td>
</tr>
</tbody>
</table>

**NHANES-III MULTIPLY IMPUTED DATA, ADULTS {20+}**

The overall contrast is displayed in *Exhibit 8*. The weighted mean body weight among respondents who consider their health to be ‘fair’ or ‘poor’ is significantly smaller (by almost 6 pounds) in adults at or below the poverty line than those above poverty.